NAG Library Routine Document

G02ANF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G02ANF computes a correlation matrix, subject to preserving a leading principal submatrix and applying the smallest relative perturbation to the remainder of the approximate input matrix.

2 Specification

```
SUBROUTINE G02ANF (G, LDG, N, K, ERRTOL, EIGTOL, X, LDX, ALPHA, ITER,      &
                   EIGMIN, NORM, IFAIL)
INTEGER            LDG, N, K, LDX, ITER, IFAIL
REAL (KIND=nag_wp) G(LDG,N), ERRTOL, EIGTOL, X(LDX,N), ALPHA, EIGMIN,      &
                   NORM
```

3 Description

G02ANF finds a correlation matrix, X, starting from an approximate correlation matrix, G, with positive definite leading principal submatrix of order k. The returned correlation matrix, X, has the following structure:

$$X = \alpha \begin{pmatrix} A & 0 \\ 0 & I \end{pmatrix} + (1 - \alpha)G$$

where A is the k by k leading principal submatrix of the input matrix G and positive definite, and $\alpha \in [0, 1]$.

G02ANF utilizes a shrinking method to find the minimum value of α such that X is positive definite with unit diagonal.

4 References

Higham N J, Strabić N and Šego V (2014) Restoring definiteness via shrinking, with an application to correlation matrices with a fixed block *MIMS EPrint 2014.54* Manchester Institute for Mathematical Sciences, The University of Manchester, UK

5 Arguments

1: G(LDG, N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: G, the initial matrix.

On exit: a symmetric matrix $\frac{1}{2}(G + G^{\mathrm{T}})$ with the diagonal set to I.

2: LDG – INTEGER *Input*

On entry: the first dimension of the array G as declared in the (sub)program from which G02ANF is called.

Constraint: LDG $\geq$ N.

3: N – INTEGER *Input*

On entry: the order of the matrix G.

Constraint: $N > 0$.

4: K – INTEGER *Input*

On entry: k, the order of the leading principal submatrix A.

Constraint: $N \geq K > 0$.

5: ERRTOL – REAL (KIND=nag_wp) *Input*

On entry: the termination tolerance for the iteration.

If ERRTOL ≤ 0, $\sqrt{\textit{machine precision}}$ is used. See Section 7 for further details.

6: EIGTOL – REAL (KIND=nag_wp) *Input*

On entry: the tolerance used in determining the definiteness of A.

If $\lambda_{\min}(A) > N \times \lambda_{\max}(A) \times \text{EIGTOL}$, where $\lambda_{\min}(A)$ and $\lambda_{\max}(A)$ denote the minimum and maximum eigenvalues of A respectively, A is positive definite.

If EIGTOL ≤ 0, ***machine precision*** is used.

7: X(LDX, N) – REAL (KIND=nag_wp) array *Output*

On exit: contains the matrix X.

8: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G02ANF is called.

Constraint: $\text{LDX} \geq N$.

9: ALPHA – REAL (KIND=nag_wp) *Output*

On exit: α.

10: ITER – INTEGER *Output*

On exit: the number of iterations taken.

11: EIGMIN – REAL (KIND=nag_wp) *Output*

On exit: the smallest eigenvalue of the leading principal submatrix A.

12: NORM – REAL (KIND=nag_wp) *Output*

On exit: the value of $\|G - X\|_F$ after the final iteration.

13: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $= \langle value \rangle$.
Constraint: N > 0.

IFAIL $= 2$

On entry, LDG $= \langle value \rangle$ and N $= \langle value \rangle$.
Constraint: LDG $\geq$ N.

IFAIL $= 3$

On entry, K $= \langle value \rangle$ and N $= \langle value \rangle$.
Constraint: N $\geq$ K > 0.

IFAIL $= 4$

On entry, LDX $= \langle value \rangle$ and N $= \langle value \rangle$.
Constraint: LDX $\geq$ N.

IFAIL $= 5$

The k by k principal leading submatrix of the initial matrix G is not positive definite.

IFAIL $= 6$

Failure to solve intermediate eigenproblem. This should not occur. Please contact NAG.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The algorithm uses a bisection method. It is terminated when the computed α is within ERRTOL of the minimum value. The positive definiteness of X is such that it can be successfully factorized with a call to F07FDF (DPOTRF).

The number of iterations taken for the bisection will be:

$$\left\lceil \log_2\left(\frac{1}{\mathrm{ERRTOL}}\right)\right\rceil.$$

8 Parallelism and Performance

G02ANF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G02ANF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

Arrays are internally allocated by G02ANF. The total size of these arrays does not exceed $2 \times n^2 + 3 \times n$ real elements. All allocated memory is freed before return of G02ANF.

10 Example

This example finds the smallest uniform perturbation α to G, such that the output is a correlation matrix and the k by k leading principal submatrix of the input is preserved,

$$G = \begin{pmatrix} 1.0000 & -0.0991 & 0.5665 & -0.5653 & -0.3441 \\ -0.0991 & 1.0000 & -0.4273 & 0.8474 & 0.4975 \\ 0.5665 & -0.4273 & 1.0000 & -0.1837 & -0.0585 \\ -0.5653 & 0.8474 & -0.1837 & 1.0000 & -0.2713 \\ -0.3441 & 0.4975 & -0.0585 & -0.2713 & 1.0000 \end{pmatrix}.$$

10.1 Program Text

```
    Program g02anfe

!     G02ANF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g02anf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: alpha, eigmin, eigtol, errtol, norm
      Integer                          :: i, ifail, iter, k, ldg, ldx, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: g(:,:), x(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'G02ANF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem sizes
      Read (nin,*) n, k

      ldg = n
      ldx = n
      Allocate (g(ldg,n),x(ldx,n))

!     Read in the matrix G
      Read (nin,*)(g(i,1:n),i=1,n)
```

```
!     Use the defaults for EIGTOL and ERRTOL
      eigtol = -1.E0_nag_wp
      errtol = -1.E0_nag_wp

!     Calculate nearest correlation matrix
      ifail = 0

      Call g02anf(g,ldg,n,k,errtol,eigtol,x,ldx,alpha,iter,eigmin,norm,ifail)

      ifail = 0
!     Display the symmetrised input matrix
      Call x04caf('General',' ',n,n,g,ldg,'Symmetrised Input Matrix G',ifail)
      Write (nout,*)

!     Display results
      ifail = 0
      Call x04caf('General',' ',n,n,x,ldx,                                   &
        'Nearest Perturbed Correlation Matrix X',ifail)
      Write (nout,*)
      Write (nout,99999) 'K:                            ', k
      Write (nout,*)
      Write (nout,99999) 'Number of iterations taken:', iter
      Write (nout,*)
      Write (nout,99998) 'ALPHA:                      ', alpha
      Write (nout,*)
      Write (nout,99998) 'Norm value:                 ', norm
      Write (nout,*)
      Write (nout,99998) 'Smallest eigenvalue of A:', eigmin

99999 Format (1X,A,I9)
99998 Format (1X,A,F16.4)

    End Program g02anfe
```

10.2 Program Data

```
G02ANF Example Program Data
  5  3                                   :: N, K
  1.0000 -0.0991  0.5665 -0.5653 -0.3441
 -0.0991  1.0000 -0.4273  0.8474  0.4975
  0.5665 -0.4273  1.0000 -0.1837 -0.0585
 -0.5653  0.8474 -0.1837  1.0000 -0.2713
 -0.3441  0.4975 -0.0585 -0.2713  1.0000 :: End of G
```

10.3 Program Results

```
 G02ANF Example Program Results

 Symmetrised Input Matrix G
          1       2       3       4       5
 1   1.0000 -0.0991  0.5665 -0.5653 -0.3441
 2  -0.0991  1.0000 -0.4273  0.8474  0.4975
 3   0.5665 -0.4273  1.0000 -0.1837 -0.0585
 4  -0.5653  0.8474 -0.1837  1.0000 -0.2713
 5  -0.3441  0.4975 -0.0585 -0.2713  1.0000

 Nearest Perturbed Correlation Matrix X
          1       2       3       4       5
 1   1.0000 -0.0991  0.5665 -0.3826 -0.2329
 2  -0.0991  1.0000 -0.4273  0.5735  0.3367
 3   0.5665 -0.4273  1.0000 -0.1243 -0.0396
 4  -0.3826  0.5735 -0.1243  1.0000 -0.1836
 5  -0.2329  0.3367 -0.0396 -0.1836  1.0000

 K:                                 3

 Number of iterations taken:       27
```

```
ALPHA:                          0.3232

Norm value:                     0.5624

Smallest eigenvalue of A:       0.3359
```